
Canal Seepage Data Pre-processing, Spatial Distribution, and Post-processing

(Use 2005 as an Example)

Step 1: Create a dBase IV file, CL05.DBF, to store monthly seepage values for each canal.

Create an Excel file (cl04.xls) with information from USBR_Canal_Seepage.xls, CNPPID_System_Seepage.xls, and NPPD_Seepage.xls. Then input it into an Access for exporting as a dBase IV file.


The file has following format:

	
	CANAL_ID
	YR_05_1
	YR_05_02
	……
	YR_05_11
	YR_05_12




800



801



802





.





.

Monthly seepage value (in acre-feet) for each canal.



.


Canal 806 was skipped.



.



.



.



825



826



827

· Create a directory, Modeling, under the RRCA2006;

· Use ArcCatalog to copy model_p, post-elwood, and one_mi_g into the Modeling directory;

· Also copy following files into the Modeling directory:


CL05.DBF


Import_oneyear.aml


Canal_monthly.aml

Dbase2Text.apr

Step 2: Run import_oneyear.aml to convert CL05.DBF into ArcInfo INFO table file crch05.dat.
· ARC: cw C:/RRCA2006/modeling

(Create a workspace)

· ARC: w C:/RRCA2006/modeling

(Go to the workspace)

· ARC: &r import_oneyear


(Run the AML file)


The resulting file crch05.dat can not be viewed by the Windows Explorer, and can be viewed by ArcCatalog. It has similar format to CL05.DBF with an additional Rowid column. 

Listing of import_oneyear.aml File

/* import_oneyear.aml

/* Import dBaseIV file into an ArcInfo Info dbase.

&severity &error &routine bailout

&type ***** Process Start: [date -ampm] *****

/* Prompt for a year to be processed.

&s year4 = [response 'Enter year (e.g. 2004)']

&s year2 = [substr %year4% 3 2]

&type %year4% %year2%

&if [exists crch%year2%.dat -info] &then

killinfo crch%year2%.dat

dbaseinfo cl%year2%.dbf crch%year2%.dat define;canal_id canal_id 3 3 i;;y;y;

&type ***** Process End: [date -ampm] *****

&return

&routine bailout

&severity &error &fail

&return &error \~

Last error : %aml$message% \~

Line No: %aml$errorline% \~

File: %aml$errorfile% \~

Thread: %aml$errorthread%

Step 3: Run canal_monthly.aml to spatially distribute canal recharge values to individual groundwater model cells (53790 cells: 165 rows by 326 columns). 


Arc: &r canal_monthly

One set of outputs are Info dBase files: cl05_1.dbf to cl05_12.dbf.

Listing of canal_monthly.aml

&severity &error &routine bailout

&sv starttime = [date -ampm]

/* Prompt for a year to be processed.

&s year4 = [response 'Enter year (e.g. 2004)']

&s year = [substr %year4% 3 2]

&type %year4% %year%  

   /*  Join the recharge polygons (canal loss areas) with the model grid cells coverage to be able to

   /*  create weighted average recharge for each model grid cell

   &type ***** Year %year% *****

   copy model_p model%year%

   identity model%year% post-elwood can%year% poly

   additem can%year%.pat can%year%.pat acres 16 16 n 3

   tables; sel can%year%.pat

   resel;nsel; resel canal_id > 0; calc acres = area * 0.00002295684

   q

   statistics can%year%.pat can_acre%year%.stat canal_id;sum acres;;n;n;

   joinitem can%year%.pat can_acre%year%.stat can%year%.pat canal_id

   &type ***** Done with identity and acres, [date -ampm] *****

   /*  Join canal recharge rates in units of ac*ft per month for canals 

   joinitem can%year%.pat crch%year%.dat can%year%.pat canal_id

   &type ***** Done with join, [date -ampm] *****

   &do countMO = 1 &to 12 &by 1


&sv month = %countMO%

      &type ***** Year %year%, Month %month% *****

      /*  Set all negative canal recharge values in the coverage to zero

      tables;sel can%year%.pat

      resel;nsel;resel YR_%year%_%month% < 0;calc YR_%year%_%month% = 0

      q

      &type ***** Done with negative values, [date -ampm] *****

      /*  Calculate input values for numerator value for weighted average recharge, which is recharge *

      /*  area = ((acre*ft / acres) * acres)

      additem can%year%.pat can%year%.pat acfta%year%_%month% 16 16 n 3

      tables;sel can%year%.pat

      resel;nsel; resel sum-acres > 0; calc acfta%year%_%month% = YR_%year%_%month% / sum-acres

      q

      additem can%year%.pat can%year%.pat rcharea%year%_%month% 16 16 n 2

      tables; sel can%year%.pat

      resel;nsel; calc rcharea%year%_%month% = acfta%year%_%month% * acres

      q

      &type ***** Done with numerator, [date -ampm] *****

      /*  Calculating numerator value for weighted average recharge statistics where all polygons within a

      /*  model grid cell have their recharge values summed.  For select gridcell...(poly1*area1)+(poly2

      /*  *area2)..+(polyn*arean)

      statistics can%year%.pat rech%year%_%month%.stat model_;sum rcharea%year%_%month%;;n;n;

      &type ***** Done with statistics, [date -ampm] *****

      /*  Join statistics info (rech%year%_%month%.stat) file back to model grid coverage

      joinitem model%year%.pat rech%year%_%month%.stat model%year%.pat model_

      grid;setcell one_mi_g;setwindow one_mi_g

      YR_%year%_%month%t = polygrid(model%year%,sum-rcharea%year%_%month%,#,#,5280)

      YR_%year%_%month% = con(isnull(YR_%year%_%month%t),0,YR_%year%_%month%t)

      q

      &type ***** Done with grid conversion, [date -ampm] *****

      /*  Create dbase files for import into model in units of feet per second

      gridpoint YR_%year%_%month% cl%year%_%month% rech_acft

      infodbase cl%year%_%month%.pat cl%year%_%month%.dbf

      &type ***** Done with export to DBASE, [date -ampm] *****

      kill YR_%year%_%month%t all

      kill cl%year%_%month% all

      killinfo rech%year%_%month%.stat

   &end   /* month count

   kill model%year% all

   killinfo can_acre%year%.stat

   kill can%year% all 

&type  ***** Process Start: %starttime% *****

&type  ***** Process End: [date -ampm] *****

&return

&routine bailout

&severity &error &fail

&return &error \~

Last error : %aml$message% \~

Line No: %aml$errorline% \~

File: %aml$errorfile% \~

Thread: %aml$errorthread%

Step 4: Run Avenue file ArcInfoDbase2Txt.ave to convert Info dbase files into text files. The resulting files are: cl05_1.txt, cl05_2.txt, …, cl05_10.txt, cl05_11.txt, and cl05_12.txt.

· Open dbase2Text.apr (ArcView project file) by double-clicking it;

· Select Tables button; delete all existing dbase files in it;

· Add those dBase files need to be converted: cl05_1.dbf to cl05_12.dbf; and

· Select Scripts button to run ArcInfoDbase2Text.ave.

Listing of ArcInfoDase2Txt.ave file:

numYear4d = msgbox.input("Enter Year","dBASE to Text File Conversion","2003").AsNumber

'for each numYear4d in 1940..2002

  for each numMonth in 1..12

    strYear2d = numYear4d.AsString.Right(2)

    strID1 = "cl" + strYear2d + "_" + numMonth.AsString + "_"

    strID2 = "cl" + strYear2d + "_" + numMonth.AsString + "_id"

    dbfFileName = "cl" + strYear2d + "_" + numMonth.AsString + ".dbf"

    txtFileName = "cl" + strYear2d + "_" + numMonth.AsString + ".txt"

    dbfVTab = VTab.Make(dbfFileName.AsFileName, false, false)

    dbfTable = Table.Make(dbfVTab)

    dbfTable.SetName(dbfFileName)

    dbfTable.GetWin.Open

    fldArea = dbfVTab.FindField("Area")

    fldArea.SetVisible(false)

    fldPerimeter = dbfVTab.FindField("Perimeter")

    fldPerimeter.SetVisible(false)

    fldID1 = dbfVTab.FindField(strID1)

    fldID1.SetVisible(false)

    fldID2 = dbfVTab.FindField(strID2)

    fldID2.SetVisible(false)

    'dbfVTab.SetEditable(true)

    'dbfVTab.RemoveFields({fldArea,fldPerimeter,fldID1,fldID2})

    'dbfVTab.SetEditable(false)

    dbfVTab.Export (txtFileName.asFileName, DText, FALSE)

  end

'end

Quality Check: 

Check monthly total seepage values against those in the resulting monthly text files.


See QualityCheck.xls file.

